LINEAR MIXED MODELS WITH REPEATED EFFECTS

Introduction and Examples Using SAS/STAT® Software

Jerry W Davis, University of Georgia, Griffin Campus. 2017

UNIVERSITY OF GEORGIA
Griffin Campus

REPEATED MEASURES EXPERIMENTS

Subjects (experimental units) are measured in time or in space.

Time – subjects receive a treatment and are monitored over time.

Spatial – root densities at increasing soil depths.

The examples use data from a measurement over time experiment, but the methods also apply to spatial experiments.

Time or spatial effects can exist in all common experimental designs.

REPEATED MEASURES EXPERIMENTS

Factors: Rep, Treatment, Time.

Treatment: *Between-subjects* factor.

Each subject receives a different treatment.

Time: *Within-subjects* factor.

For each treatment, subjects are measured at multiple time points.

REPEATED MEASURES EXPERIMENTS

Between-subject factor	
Rep 1	
Treatment 5	Treatment 6
Treatment 4	Treatment 1
Treatment 8	Treatment 2
Treatment 7	Treatment 3

Within-subject factor	
DAP	Treatment 5
15	3.43
45	1.98
60	1.11

REPEATED MEASURES EXPERIMENTS

Between-subject factors

Randomized between subjects.
Residual errors are not correlated.

Within-subject factors

Cannot be randomized.
Residual errors may be correlated.
Correlation may depend on the difference between time points.
Violates the assumption that residual errors are independently and identically distributed (iid).

REPEATED MEASURES EXPERIMENTS

PROC GLM – Within Time

```
proc glm;
    by time;
    class rep treatment;
    model response = rep treatment;
    means treatment / lsd lines;
run;
```

REPEATED MEASURES EXPERIMENTS

PROC GLM – Split Plot in Time

```
proc glm;
    class rep treatment time;
    model response = rep treatment rep*treatment time;
    test h=treatment e=rep*treatment;
    means treatment / lsd lines e=rep*treatment;
    means time / lsd lines;
run;
```

REPEATED MEASURES EXPERIMENTS

PROC MIXED

- REP – Random Effect
- TREATMENT – Fixed Effect
- TIME – Fixed and Repeated Effect

REPEATED MEASURES EXPERIMENTS

PROC MIXED

```
proc mixed;
    class rep treatment time;
    model response = treatment time treatment*time / ddfm=kr2;
    random intercept / subject=rep;
    repeated time / type=cs subject=rep*treatment;
    lsmeans treatment time;
    store cs1;
run;
```

REPEATED MEASURES EXPERIMENTS

Subjects = Experimental units (in general)

Repeated effects should represent actual time or spatial values.

Time – days after planting, days after treatment, Julian day, etc.

Spatial – inches, feet, meters, furlongs, etc.

Do Not use coded values, like 1, 2, 3 for weeks 1, 3, 7, etc.

Misrepresents the time or spatial relationship between measurements.

REPEATED MEASURES EXPERIMENTS

Covariance Structures (commonly used for agriculture experiments)

Structure	Description	Measurement Interval
VC	Variance Components	Unequal or equally spaced
CS	Compound Symmetry	Unequal or equally spaced
AR(1)	First-Order Autoregressive	Equally spaced
SP(pow)	Spatial Power	Unequal or equally spaced
UN	Unstructured	Unequal or equally spaced

REPEATED MEASURES EXPERIMENTS

Covariance structure characteristics

VC – uncorrelated errors, like PROC GLM. Not very useful.
CS – correlation is non-zero and equal for all within-subject measurements.
AR(1) – time series, correlation decreases as the lag between measurements increases.
SP(pow) – spatial structure, similar to AR(1), but does not require equally spaced measurements.
UN – very flexible, it estimates a parameter for each measurement period. Fitting the model may be difficult because of the many parameter estimates.

SAS program

RepeatedCS_UN1.sas

REPEATED MEASURES EXPERIMENTS

Selecting the model with the best covariance structure

1. Run analyses using appropriate covariance structures.
2. Compare the AICC, AIC or BIC values (AICC is probably best).
3. Select the covariance structure that has the smaller AICC value.
4. If the values are similar, select the simpler structure or the one that makes sense within the context of the research.
5. Fit the model, make the desired tests and interpret the results.

REPEATED MEASURES EXPERIMENTS

A couple points to remember:

Selecting the covariance structure is only an intermediate step for testing the means of the fixed effects.

Remember that the information criterion values do not indicate how well a model fits the data; it is the difference between values that indicate which model provides the better fit.

REPEATED MEASURES EXPERIMENTS

Information criterion from the analyses

Structure	Type	Covariance estimate	AICC	AIC	BIC
Compound Symmetry	CS	-0.0008	49.7	49.3	47.5
First-order Autoregressive	AR(1)	0.084	49.4	49.1	47.2
Spatial (power)	SP(pow)	0.906	49.4	49.1	47.2
Unstructured	UN	*NA*	51.7	49.9	45.2

Since the AICC values are similar, CS is my choice for the covariance structure because it is the simplest.

REPEATED MEASURES EXPERIMENTS

The ANOVA table with CS as the covariance structure

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
treatment	7	21	3.47	0.0125
dap	2	48	396.07	<.0001
treatment*dap	14	48	4.28	<.0001

A significant interaction indicates the means for treatment should be tested within each level of dap (days after planting).

REPEATED MEASURES ANALYSIS

The final analysis:

```
proc mixed;
    class rep treatment dap;
    model response = treatment dap treatment*dap / ddfm=kr2;
    random intercept / subject=rep;
    repeated dap / type=cs subject=rep*treatment;
    store cs1;
run;
proc plm restore=cs1;
    lsmeans treatment dap / lines;
    slice treatment*dap / sliceby=dap / lines;
run;
```

SAS program

RepeatedCS1.sas

Jerry W Davis, University of Georgia, Griffin Campus. 2017

UNIVERSITY OF GEORGIA
Griffin Campus

REPEATED MEASURES EXPERIMENTS

GLIMMIX has different syntax

```
random dap / residual type=cs subject=rep*treatment;
```

GLIMMIX also has a statement to test parts of a model.

```
covtest 'is GLM OK?' glm;
```

The result: (Pr > ChiSq) = 0.0052 means that we reject the null hypothesis that a model fit by PROC GLM is as good as the repeated measures model.

REPEATED MEASURES EXPERIMENTS

Do not panic!

- Modeling repeated measures with MIXED or GLIMMIX is not hard.
  - Add the **repeated** or **random** statement with correct syntax.
  - Did the procedure converge?
  - Do the results look right? Compare to GLM results.
  - Using a non-optimal covariance structure is seldom critical.

Correlated within-subject measurements has been addressed, which should satisfy colleagues and manuscript reviewers.

Convergence

Jerry W Davis, University of Georgia, Griffin Campus. 2017

Questions?

jwd@uga.edu

770 228-7237

UNIVERSITY OF GEORGIA
Griffin Campus